# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: Fernando Jiménez Solano <fjim@sdfeu.org>
# Contributor: Sven-Hendrik Haase <sh@lutzhaase.com>

pkgname=python-pyacoustid
pkgver=1.3.0
pkgrel=1
pkgdesc='Bindings for Chromaprint acoustic fingerprinting and the Acoustid API'
arch=(any)
url=https://github.com/beetbox/pyacoustid
license=(MIT)
depends=(
  chromaprint
  python
)
makedepends=(
  git
  python-setuptools
)
_tag=d2698e46aa23b9d3380c7b97278fc84fac502d7f
source=(git+https://github.com/beetbox/pyacoustid.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd pyacoustid

  git describe --tags | sed 's/^v//'
}

build() {
  cd pyacoustid

  python setup.py build
}

package() {
  cd pyacoustid

  python setup.py install --root="${pkgdir}" --optimize=1 --skip-build
  install -Dm 644 LICENSE -t "${pkgdir}"/usr/share/licenses/python-pyacoustid/
}

# vim: ts=2 sw=2 et:
